类 资源管理 : 窗口

	@布局配置([[根布局=真,宽度=-1,高度=-1,背景颜色=-1576716,纵向布局=真]])
	变量 线性布局1 : 线性布局
	@布局配置([[父布局=线性布局1,内容="资源管理",粗体=真,字体大小="16",字体颜色=-16777216,宽度="-1",高度="60dp",对齐方式=17]])
	变量 文本框1 : 文本框
	@布局配置([[父布局=线性布局1,宽度="-1",高度="40dp",@下外边距="10dp"]])
	变量 线性布局2 : 线性布局
	@布局配置([[父布局=线性布局2,宽度="-1",高度="-1",圆角=24,卡片阴影=0,@左外边距="15dp",@右外边距="15dp"]])
	变量 卡片布局1 : 卡片布局
	@布局配置([[父布局=卡片布局1,宽度="-1",高度="-1"]])
	变量 相对布局1 : 相对布局
	@布局配置([[父布局=相对布局1,宽度="1dp",高度="-1",@位于某组件左边=1,卡片阴影=0,背景颜色=-1579547,@上外边距="10dp",@下外边距="10dp",ID=2]])
	变量 卡片布局2 : 卡片布局
	@布局配置([[父布局=相对布局1,宽度="80dp",高度="-1",ID=1,@位于布局右边=真,对齐方式=17,内容="搜索",粗体=真,字体颜色=-16777216]])
	变量 按钮_搜索 : 文本框
	@布局配置([[父布局=相对布局1,宽度="-1",高度="-1",@位于某组件左边=2,背景颜色=16777215,单行显示=真,单行输入=真,提示文本="请输入关联词",内容="",@左外边距="10dp",对齐方式=16,字体颜色=-16777216,@右外边距="10dp",字体大小="14"]])
	变量 编辑框1 : 编辑框
	@布局配置([[父布局=线性布局1,宽度="-1",高度="-1",@下外边距="15dp"]])
	变量 线性布局3 : 线性布局
	@布局配置([[父布局=线性布局3,宽度="-1",高度="-1",显示滑块条=假,始终显示快速滚动条=假,列数=1]])
	变量 自定义宫格列表框1 : 自定义宫格列表框
	@布局配置([[父布局=线性布局1,宽度="-1",高度="-1",对齐方式=17,可视=假]])
	变量 线性布局4 : 线性布局
	@布局配置([[父布局=线性布局4,宽度="150dp",高度="150dp",纵向偏移="-50dp"]])
	变量 Lottie加载库1 : Lottie加载库

	变量 当前数组:JSON数组
	变量 审核状态集合:文本集合
	变量 蓝莓接口1:蓝莓接口
	事件 资源管理:创建完毕()
		本对象.显示标题栏 = 假
		本对象.屏幕方向 = 1
		本对象.状态栏颜色 = 文本到颜色值("#E7F0F4")
		本对象.导航栏颜色 = 文本到颜色值("#E7F0F4")
		本对象.状态栏字体黑色 = 真
		弹出等待("正在加载……")
		蓝莓接口1.获取所有资源资源()
	结束 事件
	
	事件 蓝莓接口1:获取所有资源结果(结果 : 文本)
		如果 结果==空 则
			线性布局4.可视 = 真
			线性布局3.可视=假
			Lottie加载库1.加载动画("Lottie/无功能列表.json")
			Lottie加载库1.是否循环播放(假)
		否则
			线性布局4.可视 = 假
			线性布局3.可视=真
			变量 结果内容:JSON对象=结果
			变量 临时数组:JSON数组=结果内容.取JSON数组("data")
			变量 临时当前数组:JSON数组
			循环(i, 临时数组.长度-1,-1,-1 )
				临时当前数组.置入(临时数组.取JSON对象(i))
				如果 临时数组.取JSON对象(i).取文本("是否已审核")=="真" 则
					审核状态集合.添加成员("已审核")
				否则
					如果 临时数组.取JSON对象(i).取文本("拒绝原因")=="空"则
						审核状态集合.添加成员("未审核")
					否则
						审核状态集合.添加成员("已拒绝")
					结束 如果
				结束 如果
			结束 循环
			当前数组=临时当前数组
			关闭等待()
			自定义宫格列表框1.更新项目(当前数组.长度)
		结束 如果
	结束 事件

	事件 编辑框1:焦点被改变(是否获得焦点 : 逻辑型)
		如果 是否获得焦点==真 则
			编辑框1.提示文本 = ""
		否则
			编辑框1.提示文本 = "请输入关键词"
		结束 如果
	结束 事件

	事件 自定义宫格列表框1:加载布局(索引 : 整数,项目布局 : 组件容器) : 组件容器
		变量 项目布局1:资源对象管理布局
		项目布局1.资源_标题.内容=当前数组.取JSON对象(索引).取文本("文件标题")
		项目布局1.资源_介绍.内容=当前数组.取JSON对象(索引).取文本("文件介绍")
		项目布局1.资源_赞.内容=当前数组.取JSON对象(索引).取文本("获赞数量")
		项目布局1.资源_下载.内容=当前数组.取JSON对象(索引).取文本("文件图标")
		项目布局1.资源_日期.内容=当前数组.取JSON对象(索引).取文本("上传时间")
		变量 发布者:JSON对象=当前数组.取JSON对象(索引).取文本("文件备注")
		Glide图片加载.加载图片(项目布局1.资源_头像,发布者.取文本("头像"))
		Glide图片加载.加载图片(项目布局1.图片框_头像框,全局配置.附加资源+"头像框/"+发布者.取文本("头像框"))
		项目布局1.资源_昵称.内容 = 发布者.取文本("昵称")
		项目布局1.资源_昵称.字体 = 全局配置.资源存储路径+"/配置文件/基础文件/字体/默认.ttf"
		变量 当前格式:文本=当前数组.取JSON对象(索引).取文本("文件格式")
		假如 当前格式
			是 "结绳-项目"
				如果 当前数组.取JSON对象(索引).取文本("文件大小")=="4.6"则
					项目布局1.资源_结绳版本.图片资源 = "../../素材/内置/png/结绳源码.png"
				否则
					项目布局1.资源_结绳版本.图片资源 = "../../素材/内置/png/结绳源码2.png"
				结束 如果
				项目布局1.资源_兼容.内容="结绳-"+当前数组.取JSON对象(索引).取文本("文件大小")
				项目布局1.资源_类型.图片资源 = "../../素材/内置/png/压缩包_结绳项目.png"
			是 "结绳-插件"
				如果 当前数组.取JSON对象(索引).取文本("文件大小")=="4.6"则
					项目布局1.资源_结绳版本.图片资源 = "../../素材/内置/png/结绳源码.png"
				否则
					项目布局1.资源_结绳版本.图片资源 = "../../素材/内置/png/结绳源码2.png"
				结束 如果
				项目布局1.资源_兼容.内容="结绳-"+当前数组.取JSON对象(索引).取文本("文件大小")
				项目布局1.资源_类型.图片资源 = "../../素材/内置/png/压缩包_结绳插件.png"
			是 "结绳-绳包"
				如果 当前数组.取JSON对象(索引).取文本("文件大小")=="4.6"则
					项目布局1.资源_结绳版本.图片资源 = "../../素材/内置/png/结绳源码.png"
				否则
					项目布局1.资源_结绳版本.图片资源 = "../../素材/内置/png/结绳源码2.png"
				结束 如果
				项目布局1.资源_兼容.内容="结绳-"+当前数组.取JSON对象(索引).取文本("文件大小")
				项目布局1.资源_类型.图片资源 = "../../素材/内置/png/压缩包_绳包大全.png"
			是 "其他资源"
				项目布局1.资源_结绳版本.图片资源 = "../../素材/内置/png/未知文件.png"
				项目布局1.资源_兼容.内容="未知文件"
				项目布局1.资源_类型.图片资源 = "../../素材/内置/png/压缩包_其他资源.png"
		结束 假如
		假如 审核状态集合[索引]
			是 "已审核"
				项目布局1.线性布局8.背景颜色 = 文本到颜色值("#14880A")
				项目布局1.图片框_审核状态.图片资源 = "../../素材/内置/png/已审核.png"
			是 "未审核"
				项目布局1.线性布局8.背景颜色 = 文本到颜色值("#F88138")
				项目布局1.图片框_审核状态.图片资源 = "../../素材/内置/png/未审核.png"
			是 "已拒绝"
				项目布局1.线性布局8.背景颜色 = 文本到颜色值("#F55353")
				项目布局1.图片框_审核状态.图片资源 = "../../素材/内置/png/已拒绝.png"
		结束 假如
		返回 项目布局1
	结束 事件

	事件 自定义宫格列表框1:项目被单击(被单击组件 : 可视化组件,索引 : 整数)
		变量 对话框1:DX消息框
		对话框1.标题(当前数组.取JSON对象(索引).取文本("文件标题"))
		对话框1.内容(当前数组.取JSON对象(索引).取文本("文件介绍"))
		对话框1.样式(DX样式.MIUI)
		对话框1.按钮3("修改资源")
		对话框1.按钮1("查看资源")
		对话框1.按钮2("删除资源")
		变量 文本属性1:DX文本属性
		文本属性1.颜色 = 黑色
		文本属性1.粗体 = 真
		对话框1.按钮3属性(文本属性1)
		对话框1.按钮2属性(文本属性1)
		对话框1.按钮方向(1)
		对话框1.显示()
		事件 对话框1:按钮1被单击() : 逻辑型
			假如 审核状态集合[索引]
				是 "已审核"
					资源详情.审核状态="已审核"
				是 "未审核"
					资源详情.审核状态="未审核"
				是"已拒绝"
					资源详情.审核状态="已拒绝"
			结束 假如
			资源详情.当前对象=当前数组.取JSON对象(索引)
			资源详情.窗口来源="资源管理"
			切换窗口(资源详情)
			返回 假
		结束 事件
		事件 对话框1:按钮3被单击() : 逻辑型
			修改资源.当前对象=当前数组.取JSON对象(索引)
			切换窗口(修改资源)
			返回 假
		结束 事件
		事件 对话框1:按钮2被单击() : 逻辑型
			对话框1.关闭()
			变量 删除提示2:DX消息框
			删除提示2.样式(DX样式.MIUI)
			删除提示2.标题("再次确认")
			删除提示2.内容("重要操作，确定删除吗？")
			删除提示2.可取消(真)
			删除提示2.按钮1("确定")
			删除提示2.按钮2("取消")
			删除提示2.按钮方向(1)
			删除提示2.显示()
			事件 删除提示2:按钮1被单击() : 逻辑型
				删除提示2.关闭()
				蓝莓接口1.删除资源(当前数组.取JSON对象(索引).取文本("id"))
				返回 假
			结束 事件
			返回 假
		结束 事件

	结束 事件

	事件 按钮_搜索:被单击()
		如果 编辑框1.内容 == "" 则
			弹出提示("请输入关键词")
		否则
			弹出等待("正在搜索……")
			蓝莓接口1.搜索资源(编辑框1.内容)
		结束 如果
	结束 事件
	事件 蓝莓接口1:搜索资源结果(结果 : 文本)
		如果 结果==空 则
			关闭等待()
			线性布局4.可视 = 真
			线性布局3.可视=假
			Lottie加载库1.加载动画("Lottie/无功能列表.json")
			Lottie加载库1.是否循环播放(假)
		否则
			线性布局4.可视 = 假
			线性布局3.可视=真
			变量 结果内容:JSON对象=结果
			变量 临时数组:JSON数组=结果内容.取JSON数组("data")
			变量 临时当前数组:JSON数组
			循环(i, 临时数组.长度-1,-1,-1 )
				临时当前数组.置入(临时数组.取JSON对象(i))
			结束 循环
			当前数组=临时当前数组
			关闭等待()
			自定义宫格列表框1.更新项目(当前数组.长度)
		结束 如果
	结束 事件

	@虚拟方法
	方法 返回键被按下()
		切换窗口(运行状态)
		播放切换动画(全局配置.从右向左_进入,全局配置.从右向左_退出)
	结束 方法
	

结束 类

类 资源对象管理布局:组件容器

	@布局配置([[根布局=真,宽度=-1,高度="150dp",背景颜色=-1576716]])
	变量 线性布局1 : 线性布局
	@布局配置([[父布局=线性布局1,宽度="-1",高度="-1",@左外边距="15dp",@右外边距="15dp",@上外边距="5dp",@下外边距="5dp",圆角=24,卡片阴影=0]])
	变量 卡片布局1 : 卡片布局
	@布局配置([[父布局=卡片布局1,宽度="-1",高度="-1",背景颜色=-1973791]])
	变量 线性布局2 : 相对布局
	@布局配置([[父布局=线性布局2,背景颜色=-15431670,宽度="10dp",高度="-1"]])
	变量 线性布局8 : 线性布局
	@布局配置([[父布局=线性布局2,宽度="-1",高度="-1",圆角=24,卡片阴影=0,@外边距="1dp",@左外边距="3dp"]])
	变量 卡片布局2 : 卡片布局
	@布局配置([[父布局=卡片布局2,宽度="-1",高度="-1",纵向布局=真]])
	变量 线性布局3 : 线性布局
	@布局配置([[父布局=线性布局3,宽度="-1",高度="40dp"]])
	变量 相对布局1 : 相对布局
	@布局配置([[父布局=相对布局1,高度="-1",对齐方式=16]])
	变量 线性布局4 : 线性布局
	@布局配置([[父布局=线性布局4,宽度="35dp",高度="35dp",@左外边距="10dp",@右外边距="5dp"]])
	变量 相对布局2 : 相对布局
	@布局配置([[父布局=相对布局2,宽度="-1",高度="-1",图片资源="../../素材/内置/png/默认头像.jpg",@外边距="4dp",边框宽度=0]])
	变量 资源_头像 : 圆形图片框
	@布局配置([[父布局=相对布局2,宽度="-1",高度="-1",图片资源="../../素材/内置/png/头像框_默认.png",拉伸方式=0]])
	变量 图片框_头像框 : 图片框
	@布局配置([[父布局=线性布局4,内容="结绳用户",单行显示=真,字体颜色=-16777216,粗体=真,字体大小="13"]])
	变量 资源_昵称 : 文本框
	@布局配置([[父布局=相对布局1,宽度="25dp",高度="25dp",图片资源="../../素材/内置/png/压缩包_结绳项目.png",@右外边距="17dp",@位于布局中间=真,@位于布局右边=真]])
	变量 资源_类型 : 图片框
	@布局配置([[父布局=线性布局3,@左外边距="23dp",高度="25dp",对齐方式=16,@上外边距="5dp"]])
	变量 线性布局5 : 线性布局
	@布局配置([[父布局=线性布局5,宽度="20dp",高度="20dp",图片资源="../../素材/内置/png/压缩包_结绳项目.png",@右外边距="5dp",可视=假]])
	变量 图片框3 : 图片框
	@布局配置([[父布局=线性布局5,内容="软件标题",单行显示=真,粗体=真,字体颜色=-16777216]])
	变量 资源_标题 : 文本框
	@布局配置([[父布局=线性布局3,内容="暂无相关介绍",单行显示=真,@左外边距="25dp",@上外边距="2dp",@下外边距="5dp",字体大小="13",字体颜色=-6710887]])
	变量 资源_介绍 : 文本框
	@布局配置([[父布局=线性布局3,宽度="-1",高度="1dp",圆角=0,卡片阴影=0,@左外边距="15dp",@右外边距="15dp",@上外边距="10dp",背景颜色=-1973791]])
	变量 卡片布局3 : 卡片布局
	@布局配置([[父布局=线性布局3,宽度="-1",高度="-1",@左外边距="20dp",@右外边距="20dp"]])
	变量 相对布局3 : 相对布局
	@布局配置([[父布局=相对布局3,高度="-1",对齐方式=16]])
	变量 线性布局6 : 线性布局
	@布局配置([[父布局=线性布局6,宽度="10dp",高度="10dp",图片资源="../../素材/内置/png/资源点赞.png"]])
	变量 图片框2 : 图片框
	@布局配置([[父布局=线性布局6,内容="0",字体大小="8",@左外边距="2dp",字体颜色=-6710887]])
	变量 资源_赞 : 文本框
	@布局配置([[父布局=线性布局6,宽度="10dp",高度="10dp",图片资源="../../素材/内置/png/资源下载.png",@左外边距="10dp"]])
	变量 图片框4 : 图片框
	@布局配置([[父布局=线性布局6,内容="0",字体大小="8",@左外边距="2dp",字体颜色=-6710887]])
	变量 资源_下载 : 文本框
	@布局配置([[父布局=线性布局6,宽度="10dp",高度="10dp",图片资源="../../素材/内置/png/资源下载.png",@左外边距="10dp",可视=假]])
	变量 图片框5 : 图片框
	@布局配置([[父布局=线性布局6,宽度="9dp",高度="9dp",图片资源="../../素材/内置/png/结绳源码2.png",@左外边距="10dp"]])
	变量 资源_结绳版本 : 圆形图片框
	@布局配置([[父布局=线性布局6,内容="结绳-4.1.8",字体大小="8",@左外边距="2dp",字体颜色=-6710887]])
	变量 资源_兼容 : 文本框
	@布局配置([[父布局=相对布局3,@位于布局右边=真,高度="-1",对齐方式=8388629]])
	变量 线性布局7 : 线性布局
	@布局配置([[父布局=线性布局7,宽度="10dp",高度="10dp",图片资源="../../素材/内置/png/日历.png",@左外边距="10dp"]])
	变量 图片框6 : 图片框
	@布局配置([[父布局=线性布局7,内容="2025年10月1日",字体大小="8",@左外边距="2dp",字体颜色=-6710887]])
	变量 资源_日期 : 文本框
	@布局配置([[父布局=线性布局2,图片资源="../../素材/内置/png/已审核.png",@位于布局中间=真,宽度="150dp",高度="-1",透明度=0.16]])
	变量 图片框_审核状态 : 图片框

结束 类